

# Построение солитонов КдФ при помощи суперпозиции преобразований Бэклунда

В.Э. Адлер · Классические интегрируемые системы · Лекция 11 · 19 апреля 2021

---

Построение  $n$ -солитонного решения уравнения КдФ (для наглядности меняем знак  $u$ )

$$u = 2 \partial_x^2 \log W(y_1, \dots, y_n), \quad y_j = e^{X_j} + (-1)^{n-j} e^{-X_j}, \quad X_j = k_j x + 4 k_j^3 t + \delta_j, \quad 0 < k_n < \dots < k_1,$$

по схеме, описанной на лекции. Если считать вронскиан непосредственно, то  $n > 4$  уже доставляет трудности. При помощи суперпозиции можно легко построить 10–12 солитонов. При большем числе и здесь счет становится медленным, а также накапливаются ошибки в арифметических операциях. Если немного постараться, то можно дойти до 15, но дальше *Mathematica* справляется плохо.

In[1]:=

```

(* ЧИСЛО СОЛИТОНОВ *)
n = 10;

(* задаем случайные собственные значения и сортируем их по убыванию,
также задаем случайные сдвиги *)
K = Sort[RandomReal[{1/2, 3/2}, n], #1 > #2 &]
d = RandomReal[{-5, 5}, n]

(* фазы *)
X[j_] := K[[j]] x + 4 K[[j]]3 t + d[[j]]

(* затравочные решения *)
q[] = 0;
Do[q[j] = -K[[j]] Tanh[X[j]], {j, n, 1, -2}]
Do[q[j] = -K[[j]] Coth[X[j]], {j, n - 1, 1, -2}]

(* формула суперпозиции *)
q[i___, j_, k_] := q[i] + (K[[j]]2 - K[[k]]2) / (q[i, j] - q[i, k])

(* вычисление решения *)
u = -2 D[q @@ Range[n], x];

(* компилированная функция, это ускоряет счет *)
cu = Compile[{{x, _Real}, {t, _Real}}, Evaluate[u]];

(* построение графика *)
cgrou[t_, xmin_, xmax_] := Plot[cu[x, t], {x, xmin, xmax},
  PlotRange -> {0, 2.1 K[[1]]2},
  AxesLabel -> {"x", "u"},
  PlotStyle -> Blue, Filling -> 0,
  PlotPoints -> 50,
  BaseStyle -> {FontSize -> 14, FontFamily -> "Times New Roman"},
  ImageSize -> 500,
  Epilog -> {Text["t = " <> ToString[t], {0.8 xmax, 1.6 K[[1]]2}, {-1, 0}]}
];

(* анимация по t *)
Manipulate[cgrou[t, -50, 50],
  {{t, 0}, -10, 10, Appearance -> "Labeled"}]

```

Out[2]= {1.38536, 1.32798, 1.30164, 1.23296, 0.856945, 0.852312, 0.822863, 0.698537, 0.621531, 0.619437}

Out[3]= {-1.98328, -0.954211, 2.62222, 1.68483, -4.67912, 3.25343, 2.4194, 1.72881, 2.68573, 0.4732}

Out[12]=

